

Jednoduchý model stability alternátoru


Description

Jednoduchý model stability alternátoru s direktivní regulací buzení a turbíny doplněný tlumením.

System Parameters

$P_{max1} = 227.87$	[MW]	Maximalni cinny vykon alternatoru v predporuchovem stavu
$P_{max2} = 29.01$	[MW]	Maximalni cinny vykon alternatoru v dobe poruchy
$P_{max3} = 205.59$	[MW]	Maximalni cinny vykon alternatoru v dobe odpojeni poruchy
$C_{asPoruchy} = 1$	[s]	Cas vzniku poruchy
$C_{asVypnuti} = 1.5$	[s]	Cas vypnuti poruchy
$T_{buz} = 0.3$	[s]	Casova konstanta regulace buzeni
$P_{t0} = 60$	[MW]	Pocatecni vykon turbiny
$T_{tur} = 10$	[s]	Casova konstanta regulace turbiny
$\omega_S = 100\pi$	[rad/s]	Synchronni uhlova rychlost soustroji
$T_m = 5$	[s]	Mechanicka casova konstanta soustroji
$S_{ng} = 100$	[MVA]	Jmenovity zdanlivy vykon alternatoru
$M_{eritkoNapeti} = 80$		
$M_{eritkoProudu} = 0.09$		
$U_{nf} = 400/\text{sqrt}(3)$		
$T_{theta0} = 0.26645$		
$X_{sumac} = 915.17$		

Model



Data

*: Jednoduchý model stability alternátoru

*SYSTEM;

Pmax1=227.87; :: [MW] Maximalni cinny vykon alternatoru v predporuchovem stavu

```

Pmax2= 29.01;    :: [MW] Maximalni cinny vykon alternatoru v dobe poruchy
Pmax3=205.59;   :: [MW] Maximalni cinny vykon alternatoru v dobe odpojeni poruchy

CasPoruchy=1;   :: [s] Cas vzniku poruchy
CasVypnuti=1.5; :: [s] Cas vypnuti poruchy

: Regulace buzeni
Tbuz=0.3;       :: [s] Casova konstanta regulace buzeni
RegBuzeni=1+(TIME>=CasPoruchy)*(1-EXP(-(TIME-CasPoruchy)/Tbuz))*1.5;
:: [-] Pomerna hodnota regulace buzeni

: Vykon alternatoru
Pg=RegBuzeni*
  ((TIME< CasPoruchy)*Pmax1*SIN(Theta)+
   (TIME>=CasPoruchy)*(TIME<CasVypnuti)*Pmax2*SIN(Theta)+
   (TIME>=CasVypnuti)*Pmax3*SIN(Theta));
:: [MW] Cinny elektricky vykon alternatoru

: Vykon turbiny
Pt0=60;        :: [MW] Pocatecni vykon turbiny
Ttur=10;       :: [s] Casova konstanta regulace turbiny
RegTurbiny=1-(TIME>=CasPoruchy)*(1-EXP(-(TIME-CasPoruchy)/Ttur))*0.9; :: [-] Pomerna hodnota
regulace turbiny
Pt=RegTurbiny*Pt0; :: [MW] Mechanicky vykon turbiny

OmegaS=100pi; ::[rad/s] Synchronni uhlova rychlost soustroji
Tm=5;         ::[s] Mechanicka casova konstanta soustroji
Sng=100;      ::[MVA] Jmenovity zdanlivy vykon alternatoru
Jmot=Tm*Sng/(OmegaS*OmegaS);

: Pohybove rovnice soustroji
Mt > J dOmega = -Pt/OmegaS;
Mg > J dOmega = Pg/OmegaS;
Setrvacnost > C dOmega = Jmot;
Tlumeni > @fan dOmega,0 / 0.2m;
Integrator > @Int dOmega,Theta;

DeltaOmega=dOmega; :: [rad/s] Relativni uhlova rychlost soustroji
ZateznyUhel=Theta; :: [rad] Zatezny uhel alternatoru

: Promenne pro vytvoreni animace
MeritkoNapeti=80;
MeritkoProudu=0.09;
Unf=400/sqrt(3);
Unfx=Unf/MeritkoNapeti; :: [-] Realna cast Unf


: Ec vychazi z predporuchovych hodnot a meni se podle regulace buzeni
Theta0=0.26645;
Xsumac = 915.17;
Ec=Pt0/3/Unf/SIN(Theta0)*RegBuzeni*Xsumac;
Ecx=Ec*COS(Theta)/MeritkoNapeti; :: [-] Realna cast E'
Ecy=Ec*SIN(Theta)/MeritkoNapeti; :: [-] Imaginarni cast E'



Xsuma=3*Ec*Unf*SIN(Theta)/Pg;
: Ix=Pg/3/Unf nebo Ix=Ec*SIN(Theta)/Xsuma
Icinny = Ec*SIN(Theta) /Xsuma/MeritkoProudu; :: [-] Cinny proud



```


```
Ijalovy=- (Ec*COS(Theta)-Unf)/Xsuma/MeritkoProudu; :: [-] Jalovy proud
```

```
*TR;  
TR 0 10;  
PRINT(1001) ZateznyUhel, DeltaOmega, Pg, Pt,  
  RegBuzeni, RegTurbiny,  
  Unfx, Ecx, Ecy, Icinny, Ijalovy;  
INIT Theta=0.26645;  
RUN;  
*END;
```









Origin

Karel Nohac, KEE, FEL, ZCU v Plzni

Last Update

December 22, 2017